/*******************************************************************************************
*
*   raylib [core] example - 2d camera
*
*   This example has been created using raylib 1.5 (www.raylib.com)
*   raylib is licensed under an unmodified zlib/libpng license (View raylib.h for details)
*
*   Copyright (c) 2016 Ramon Santamaria (@raysan5)
*
********************************************************************************************/

using raylib_beef;
using raylib_beef.Types;

namespace raylib_beef_examples.Core
{
	class Camera2d : Raylib
	{
		public static void Main()
		{
			let maxBuildings = 100;

			let screenWidth = 800;
			let screenHeight = 600;


			InitWindow(screenWidth, screenHeight, "raylib [core] example - 2d camera");
			defer CloseWindow();

			Rectangle player = .(400, 280, 40, 40);
			Rectangle[maxBuildings] buildings;
			Color[maxBuildings] buildColors;
			int spacing = 0;

			for (int i < maxBuildings)
			{
				buildings[i].width = GetRandomValue(50, 200);
				buildings[i].height = GetRandomValue(100, 800);
				buildings[i].y = screenHeight - 130 - buildings[i].height;
				buildings[i].x = -6000 + spacing;

				spacing += (int)buildings[i].width;

				buildColors[i] = Color((uint8)GetRandomValue(200, 240), (uint8)GetRandomValue(200, 240), (uint8)GetRandomValue(200, 240), 255);
			}

			Camera2D camera;
			camera.target = Vector2(player.x + 20, player.y + 20);
			camera.offset = Vector2(screenWidth / 2, screenHeight / 2);
			camera.rotation = 0.0f;
			camera.zoom = 1.0f;

			SetTargetFPS(60);

			while (!WindowShouldClose())
			{
				Update:
				{
					// Player movement
					if (IsKeyDown(.KEY_RIGHT)) player.x += 2;
					else if (IsKeyDown(.KEY_LEFT)) player.x -= 2;

					// Camera target follows player
					camera.target = Vector2(player.x + 20, player.y + 20);

					// Camera rotation controls
					if (IsKeyDown(.KEY_A)) camera.rotation--;
					else if (IsKeyDown(.KEY_S)) camera.rotation++;

					// Limit camera rotation to 80 degrees (-40 to 40)
					if (camera.rotation > 40) camera.rotation = 40;
					else if (camera.rotation < -40) camera.rotation = -40;

					// Camera zoom controls
					camera.zoom += ((float)GetMouseWheelMove() * 0.05f);

					if (camera.zoom > 3.0f) camera.zoom = 3.0f;
					else if (camera.zoom < 0.1f) camera.zoom = 0.1f;

					// Camera reset (zoom and rotation)
					if (IsKeyPressed(.KEY_R))
					{
						camera.zoom = 1.0f;
						camera.rotation = 0.0f;
					}
				}


				Draw:
				{
					BeginDrawing();
					defer EndDrawing();

					ClearBackground(.RAYWHITE);

					BeginMode2D(camera);

					DrawRectangle(-6000, 320, 13000, 8000, .DARKGRAY);

					for (int i = 0; i < maxBuildings; i++) DrawRectangleRec(buildings[i], buildColors[i]);

					DrawRectangleRec(player, .RED);

					DrawLine((int32)camera.target.x, -screenHeight * 10, (int32)camera.target.x, screenHeight * 10, .GREEN);
					DrawLine(-screenWidth * 10, (int32)camera.target.y, screenWidth * 10, (int32)camera.target.y, .GREEN);

					EndMode2D();

					DrawText("SCREEN AREA", 640, 10, 20, .RED);

					DrawRectangle(0, 0, screenWidth, 5, .RED);
					DrawRectangle(0, 5, 5, screenHeight - 10, .RED);
					DrawRectangle(screenWidth - 5, 5, 5, screenHeight - 10, .RED);
					DrawRectangle(0, screenHeight - 5, screenWidth, 5, .RED);

					// DrawRectangle(10, 10, 250, 113, ColorAlpha(.SKYBLUE, 0.5f));
					DrawRectangle(10, 10, 250, 113, .SKYBLUE);
					DrawRectangleLines(10, 10, 250, 113, .BLUE);

					DrawText("Free 2d camera controls:", 20, 20, 10, .BLACK);
					DrawText("- Right/Left to move Offset", 40, 40, 10, .DARKGRAY);
					DrawText("- Mouse Wheel to Zoom in-out", 40, 60, 10, .DARKGRAY);
					DrawText("- A / S to Rotate", 40, 80, 10, .DARKGRAY);
					DrawText("- R to reset Zoom and Rotation", 40, 100, 10, .DARKGRAY);
				}
			}
		}
	}
}
